HW 1 - ASTR503

Daniel George - dgeorge5@illinois.edu

Setting working directory and constants

In[3]:=
```
SetDirectory["C:\\Users\\dan7g\\Google Drive\\Acads\\ASTR503\\"];
c = QuantityMagnitude["SpeedOfLight", "m/s"];
```

Q1) Electromagnetic Spectrum

Fetching frequencies from Wolfram Knowledgebase

In[5]:=
```
{freqs, names} = Entity["FrequencyAllocation"] /@ {"Frequencies", "Name"};
```

Cleaning up the data

In[6]:=
```
fData = Sort[MapAt[Log10@QuantityMagnitude[1. #, "Hz"] &, DeleteMissing[
      Transpose@{freqs, names}, 1, 1], {All, 1}], #1[[1, 1, 1, 1]] < #2[[1, 1, 1, 1]] &];
```

Plotting

In[22]:=
```
cls = ColorData["Rainbow"] /@ (Range[0, 6] / 6); pC = 1;
TimelinePlot[
 {Labeled[#[[1, 1]], Style[#[[2]], Bold], {Before, After}[[Mod[pC++, 2] + 1]]],
    Sequence @@ #[[1, 2 ;;]]} & /@ fData, ImageSize → 550,
 AspectRatio → .8, PlotTheme → "Web", Frame → True, FrameLabel →
  {{None, None}, {Style["Frequency (hertz)", Bold], Style["Wavelength (meters)", Bold]}},
 PlotLayout → "Overlapped", Spacings → {30, {20}}, Filling → Bottom,
 FillingStyle → Opacity[.35], PlotStyle →
  cls[[LengthWhile[{∞, 5*^19, 5.5*^16, 8*^14, 4*^14, 3*^11, 10^9, 0} - 10^#, # > 0 &] & /@
    Mean /@ Apply[Mean, fData[[All, 1, All]], {2}]]], GridLines → Automatic,
 PlotLegends → Placed[SwatchLegend[##, LegendMarkerSize → 15] &@@
    {cls, Style @@@ Transpose[{{"Gamma", "X-ray", "Ultraviolet",
         "Visible", "Infrared", "Microwave", "Radio"}, cls}]}, {.2, .65}],
 FrameTicks → {{}, {{#, "10"^ToString@#} & /@ Range[0, 20, 2],
    {Log10[c / 10^#], "10"^ToString@#} & /@ (Range[-12, 9, 2])}},
 Epilog → {Text[Style[10^ToString@#], {.434 Log[2.418 × 10^(14 + #)], 1298}, {0, 0}] & /@
    Range[-14, 7, 2], Text[Style["Energy (eV)", Bold], {-2, 1294}, {0, 0}]}]
```

Out[23]=



Q2) Transmission Curves

Importing data and rescaling

In[9]:=
```
fScale =
  Function[{p, s}, Flatten /@ Transpose[{#[[All, 1]]^p s, Rescale@#[[All, 2 ;;]]}] &];
{zm1610, nq1610} = fScale[1, 10.^-6] /@
   (Import[#, "Table"] & /@ {"mktrans_zm_16_10.dat", "mktrans_nq_16_10.dat"});
submm = fScale[-1, c / 10.^9] @Select[Import["submm.dat", "Table"][[All, {1, 3}]],
    #[[2]] > 10.^-13 &];
deCAM = fScale[1, 10.^-9] @Import["DECam.txt", "Table"][[2 ;;]];
hPACS = fScale[1, 10.^-6] @Import["Herschel_PACS" <> # <> ".txt", "Table"][[3 ;;]] & /@
   {"70", "100", "160"};
iras = fScale[1, 10.^-6] @Import["IRAS" <> # <> ".txt", "Table"][[4 ;;]] & /@
   {"25", "60", "100"};
spitzer = fScale[1, 10.^-6] @Import["Spitzer_" <> # <> ".txt", "Table"] & /@
   {"16", "24", "70"};
irac = fScale[1, 10.^-6] @Import["irac_ch" <> # <> "trans_full.txt", "Table"][[4 ;;]] & /@
   {"1", "2"};
deCAMn = # <> "-DECam" & /@ Import["DECam.txt", "Table"][[1, 3 ;;]];
```

Cleaning up noise in the datasets

In[18]:=
```
sData = MovingAverage[#, Length@# / 10000 // Ceiling] & /@
    {zm1610, nq1610, submm, Join @@ hPACS, Join @@ iras, Join @@ irac, Sequence @@ spitzer,
     Join @@ (deCAM[[All, {1, #}]] & /@ Range[2, Length@deCAM[[1]]])};
```

Plotting transmission curves

In[25]:=
```
Show[ListLogLinearPlot[sData[[1 ;; 3]], ##, PlotStyle → Opacity[.4],
    PlotLegends → Placed[{"Near-IR", "Mid-IR" , "Sub-mm"}, Top],
    FrameLabel → {"Wavelength (meters)", "Transmission  coefficient "}],
   ListLogLinearPlot[sData[[4 ;;]], ##, Filling → Bottom, PlotTheme → "Scientific",
    PlotLegends → Placed[SwatchLegend[{"Herschel", "IRAS", "IRAC",
        "Spitzer", (*Sequence@@deCAMn*)"DECam"}], Below]] & @@
  {AspectRatio → .6, PlotRange → {{2.7 × 10^-7, .001}, All}, ImageSize → 650,
   Frame → True, Joined → True}
```

Out[25]=