HW 4 - ASTR540

Daniel George- dgeorge5@illinois.edu

Q1

a)

Given equation

In[174]:=
```
eqPr = P -> -1/3 Egr / V
```

Out[174]= $P \to -\frac{Egr}{3\,V}$

Solving for R given M and ρ

In[175]:=
```
sR =
  R → Assuming[M / ρ > 0, Abs@ToRadicals@Solve[M/(4/3 Pi R^3) == ρ, R, Reals][[1, 1, 2]] // Simplify]
```

Out[175]= $R \to \frac{\left(\frac{3}{\pi}\right)^{1/3}\left(\frac{M}{\rho}\right)^{1/3}}{2^{2/3}}$

Substituting for R in the first equation

In[193]:=
```
sP = eqPr //. {Egr → -G M^2/R, V → 4/3 Pi R^3, sR} // ToRadicals
```

Out[193]= $P \to \frac{2^{2/3}\,G\,M^2\left(\frac{\pi}{3}\right)^{1/3}}{3\left(\frac{M}{\rho}\right)^{4/3}}$

This is equivalent to the given expression.

b)

Given that the pressures are equal

Out[170]= $\frac{a\,T^4}{3} = \frac{k\,\rho\,T}{m}$

Solving for T

```
In[177]:= sT = Solve[1/3 a T^4 == ρ k T/m, T][[3, 1]]
```

Out[177]= $T \to \frac{3^{1/3} k^{1/3} \rho^{1/3}}{a^{1/3} m^{1/3}}$

Substituting T in the total pressure

```
In[198]:= sP2 = P -> 2×1/3 a T^4 /. sT
```

Out[198]= $P \to \frac{2 \times 3^{1/3} k^{4/3} \rho^{4/3}}{a^{1/3} m^{4/3}}$

This is equal to the required expression.

---

c)

Equating the two pressures

```
In[201]:= eqM = Equal @@ (P /. {{sP}, {sP2}})
```

Out[201]= $\frac{2^{2/3} G M^2 \left(\frac{\pi}{3}\right)^{1/3}}{3 \left(\frac{M}{\rho}\right)^{4/3}} == \frac{2 \times 3^{1/3} k^{4/3} \rho^{4/3}}{a^{1/3} m^{4/3}}$

Solving for M

```
In[255]:= sM = Solve[eqM, M][[1, 1]] // FullSimplify
```

Out[255]= $M \to 9 \sqrt{\frac{6}{\pi}} \left(\frac{k^{4/3}}{a^{1/3} G m^{4/3} \rho^{2/3}}\right)^{3/2} \rho$

Substituting numerical values

```
In[209]:= UnitConvert[
   sM[[2]] /. {k -> k, m → 1/2 mp, G -> G, a -> a, ρ → Sun (star) ["Density"]}, "SolarMass"]
```

Out[209]= 113.7 $M_\odot$

Q2

---

a)

Formula for mean mass

```
In[222]:= eqμ = 1/μ == mH*(xC/mC + xe/me);
```

Substituting numbers

In[232]:=
```
eqμ //. {mH -> 1 m_p , me -> 1 m_e , xC → 1, xe → 6 m_e /mC, mC -> 12 m_p }
```

Out[232]= $\frac{1}{\mu} == \frac{7}{12}$

This gives the required mean mass of 12/7.

---

b)

Pressure from the virial theorem

In[298]:=
```
eqPv = P == 1/3 G M^2/ (R M / ρ)
```

Out[298]= $P == \frac{G M \rho}{3 R}$

Equating it to thermal pressure

In[245]:=
```
eqR = eqPv //. {P -> ρ k T / m, ρ → M/ (4/3 Pi R^3)}
```

Out[245]= $\frac{3 k M T}{4 m \pi R^3} == \frac{G M^2}{4 \pi R^4}$

Solving for R

In[296]:=
```
sR2 = Solve[eqR, R][[1, 1]]
```

Out[296]= $R \to \frac{G m M}{3 k T}$

Ratio of radius with respect to Sun

In[305]:=
```
rS = Divide @@
  (sR2[[2]] /. {G -> G , k -> k } /. {{M → 10 Sun (star) ["Mass"], m → 12/7 m_p , T -> 6 × 10^8 K },
    {M → Sun (star) ["Mass"], m → 1/2 m_p , T -> 1.5 × 10^7 K }})
```

Out[305]= 0.85714285714286

---

c)

Finding temperature given luminosity

In[314]:=
```
Solve[ 10^7 L_⊙ == 4 Pi (rS 1 R_⊙)^2 σ T^4, T, Reals][[2, 1]]
```

Out[314]= T → 350 617.28036314 K

d)

Fraction of energy released

In[321]:=
```
ef = (24 - 23.985) / 24
```
Out[321]= 0.000625000000000002

e)

Time taken to deplete 10% of carbon

Energy output divided by luminosity:

In[325]:=
```
UnitConvert[10 M☉ * .1 * ef c ^2 / 1 × 10^7 L☉N , "Years"]
```
Out[325]= 920.43081892875 yr

Q3

a)

Defining f(E)

In[180]:=
```
f[En_] = Exp[-En / (k T) - Sqrt[EG / En]]
```
Out[180]= $e^{-\sqrt{\frac{EG}{En}}-\frac{En}{k\,T}}$

Finding maximum E_0

In[181]:=
```
E0 = Solve[D[f[En], En] == 0, En][[1, 1, 2]]
```
Out[181]= $\frac{EG^{1/3}\,k^{2/3}\,T^{2/3}}{2^{2/3}}$

b)

Taylor expansion around E_0 to 2nd order

In[182]:=
```
fA = Series[Log@f[En], {En, E0, 2}] // Normal // PowerExpand // Exp
```
Out[182]= $e^{-\frac{3\left(En-\frac{EG^{1/3}\,k^{2/3}\,T^{2/3}}{2^{2/3}}\right)^2}{2\times 2^{1/3}\,EG^{1/3}\,k^{5/3}\,T^{5/3}}-\frac{EG^{1/3}}{2^{2/3}\,k^{1/3}\,T^{1/3}}-\frac{2^{1/3}\,EG^{1/3}}{k^{1/3}\,T^{1/3}}}$

Finding standard deviation

In[183]:=
```
sσ = Solve[-1/(2 σ^2) == Last@CoefficientList[Log@fA // PowerExpand, En], σ][[2, 1]]
```

Out[183]= $\sigma \to \frac{2^{1/6}\,EG^{1/6}\,k^{5/6}\,T^{5/6}}{\sqrt{3}}$

---

c)

Integrating the approximation for f(E)

In[184]:=
```
int = Integrate[fA, {En, -∞, ∞}, Assumptions → {k > 0, T > 0, EG > 0}]
```

Out[184]= $\frac{1}{\sqrt{3}} 2^{2/3}\, e^{-\frac{3\left(\frac{EG}{k\,T}\right)^{1/3}}{2^{2/3}}} \sqrt{\pi}\,\left(EG\,k^5\,T^5\right)^{1/6}$

Checking whether this is equal to RHS

In[185]:=
```
int == Sqrt[2 Pi] f[E0] σ /. sσ // PowerExpand // Simplify
```

Out[185]= True

Therefore the integral is equivalent to the given expression.

Q4

Nuclear power density

In[186]:= $\epsilon[T\_] = \left(2^{\wedge}(5/3)\,\sqrt{2}\,e^{-\frac{3\left(\frac{EG}{k\,T}\right)^{1/3}}{2^{2/3}}}\,Q\,\rho\,EnG^{1/6}\,S0\,xA\,xB\right) \Big/ \left(\sqrt{3}\,(k\,T)^{2/3}\,\sqrt{\mu}\,aA\,aB\,mH^{\wedge}2\right);$

$\partial\log(\epsilon)/\partial\log(T)$

In[265]:=
```
sβ = β -> D[Log@ϵ[T], T] / D[Log[T], T] /. T → T0 // PowerExpand // Simplify
```

Out[265]= $\beta \to -\frac{2}{3} + \frac{EG^{1/3}}{2^{2/3}\,k^{1/3}\,T0^{1/3}}$

This is the required value of *β*.

Numerical value for p + p

In[269]:= $s\beta$ /. $\{T0 \to 1.57\times10^{7}\,\mathrm{K},\ k \to k,\ EG \to 500\,\mathrm{keV}\}$

Out[269]= $\beta \to 3.854104659328$

Numerical value for p + C

In[270]:=
```
sβ /. {T0 → 1.57 × 10^7 K , k -> k , EG -> 35.5 MeV }
```
Out[270]= $\beta \to 18.053023481094$

Q5

---

a)

Defining E(v) and f(v)

In[188]:=
```
En[v_] := -ψ[r] - 1/2 v^2
f[v_] := If[En[v] > 0, F En[v]^(n - 3/2), 0]
```

Integrating f(v) over all v

In[190]:=
```
sρ = ρ[r] -> Integrate[4 Pi f[v] v^2, {v, 0, Sqrt[-2 ψ[r]]}, Assumptions → ψ[r] < 0 && n > 1/2]
```
Out[190]= $\rho[r] \to \left(2\sqrt{2}\ F\ \pi^{3/2}\ \mathrm{Gamma}\left[-\frac{1}{2}+n\right]\ (-\psi[r])^n\right) \Big/ \mathrm{Gamma}[1+n]$

---

b)

Substituting ψ in the Poisson equation

In[191]:=
```
eqP = Laplacian[ψ[r], {r, θ, ϕ}, "Spherical"] == 4 Pi G ρ[r] /. sρ;
eqP // TraditionalForm
```
Out[191]//TraditionalForm=

$$\psi''(r) + \frac{2\,\psi'(r)}{r} = \frac{8\sqrt{2}\ \pi^{5/2}\,F\,G\,\Gamma\left(n-\frac{1}{2}\right)(-\psi(r))^n}{\Gamma(n+1)}$$

Lane-Emden equation

In[192]:=
```
eqLE = 1/ξ^2 D[ξ^2 D[θ[ξ], ξ], ξ] + θ[ξ]^n == 0 // Simplify; eqLE // TraditionalForm
```
Out[192]//TraditionalForm=

$$\theta''(\xi) + \frac{2\,\theta'(\xi)}{\xi} + \theta(\xi)^n = 0$$

Thus we can see that this differential equation is the same as the Lane-Emden equation with the variables scaled appropriately as follows:

$\psi \to C^{\frac{1}{n}}\ \theta$

Out[264]= $r \to \sqrt{C^{\frac{1}{n}}}\ \xi$

$C \to \frac{(-1)^{-n}\ \mathrm{Gamma}[1+n]}{8\sqrt{2}\ F\ G\ \pi^{5/2}\ \mathrm{Gamma}\left[-\frac{1}{2}+n\right]}$

c)

Substituting $A \Big/ \sqrt{\frac{r^2}{R0^2}+1}$ for ψ(r)

In[259]:=
```
eqA = eqP /. n → 5 /. ψ → (A / Sqrt[1 + (# / R0)^2] &)
```

Out[259]=
$$\frac{3\,A\,r^2}{\left(1+\frac{r^2}{R0^2}\right)^{5/2} R0^4} - \frac{3\,A}{\left(1+\frac{r^2}{R0^2}\right)^{3/2} R0^2} == -\frac{7\,A^5\,F\,G\,\pi^3}{8\sqrt{2}\left(1+\frac{r^2}{R0^2}\right)^{5/2}}$$

Checking whether a solution exists

In[275]:=
```
sA = Solve[eqA, A][[-1]][[1]] // Simplify
```

Out[275]=
$$A \to \frac{\left(\frac{3}{7}\right)^{1/4} 2^{7/8} \left(1+\frac{r^2}{R0^2}\right)^{1/4}}{\pi^{3/4}\left(F\,G\left(r^2+R0^2\right)\right)^{1/4}}$$

Therefore for the above value of A, a solution of this form exists.

Computing density

In[282]:=
```
sρ =
  ρ[r] -> Laplacian[ψ[r], {r, θ, ϕ}, "Spherical"] / (4 Pi G) /. ψ → (A / Sqrt[1 + (# / R0)^2] &) /. sA //
   FullSimplify
```

Out[282]=
$$\rho[r] \to -\left(\left(3\left(\frac{3}{7}\right)^{1/4} F^2\,G\,R0^2\right) \Big/ \left(2\times 2^{1/8}\,\pi^{7/4}\left(1+\frac{r^2}{R0^2}\right)^{1/4}\left(F\,G\left(r^2+R0^2\right)\right)^{9/4}\right)\right)$$

Asymptotic limit

Using Taylor expansion for the reciprocal of r:

In[326]:=
```
Normal@Series[ρ[r] /. sρ /. r → 1 / u, {u, 0, 6}] /. u → 1 / r // Simplify
```

Out[326]=
$$-\frac{3\left(\frac{3}{7}\right)^{1/4} F\left(\frac{1}{R0^2}\right)^{3/4} R0^4}{2\times 2^{1/8}\left(F\,G\right)^{5/4}\pi^{7/4}\,r^5}$$

Thus we can see that the density scales as r^{-5}.