HW 3 - ASTR404

Daniel George - dgeorge5@illinois.edu

Q5)

a)

Mean free path

The mean free path λ is given by:

$$\lambda = \frac{1}{\kappa\,\rho}$$

Substituting values at the center of the sun:

```
λ → 1/(.217 m²/kg 1.53 × 10^5 kg/m³)
```

```
λ → 0.000030119574711605 m
```

b)

Using a Monte Carlo random walk simulation

Distance traveled in time T averaged over 500 runs

```
d[T_] := With[{λ = 0.0000301195747116, s = 500, c = 299792458},
  ParallelSum[Norm@Total[λ RandomPoint[Sphere[], T c / λ // Floor]], s] / s]
```

Computing distances traveled in different times

```
data = {#, d@#} & /@ (10.^Range[-11, -7, .1]);
```

Plotting distance vs time and the best fit line

```
Print["Meters traveled in T seconds is approximately: "];
fit = E^LinearModelFit[Log@data, {x}, x][Log@T] // Echo;
Show[ListLogLogPlot[data, Frame → True, PlotRange → All,
  ImageSize → 500, FrameLabel → {"Time (seconds)", "Distance (meter)"}],
 LogLogPlot[Evaluate@fit, {T, 10^-13, 10^-5}, PlotStyle → {Thin, Red}]]
```

```
Meters traveled in T seconds is approximately:
```

» $87.684466405701\, T^{0.50027607989649}$



A good guess for the distance formula

```
d == FindFormula[data][T]
```

$d == 87.057750474536\sqrt{T}$

Extrapolating the formula to find time to reach surface

```
UnitConvert[
 Solve[QuantityMagnitude["SolarRadius", "Meters"] == FindFormula[data][T], T][[1, 1, 2]] s,
 "Years"]
```

$2.0238241764831 \times 10^{6}$ yr

Therefore the average time required to reach the surface is about 2 million years.

Using analytic formula for random walk in 3D

The analytic formula for the mean distance in d-dimensions for a random-walk after N steps is:

$$\textbf{mean distance} = \frac{\sqrt{\frac{2N}{d}}\,\Gamma\left(\frac{d+1}{2}\right)}{\Gamma\left(\frac{d}{2}\right)}$$

Plugging in numbers for our problem:

```
dT = Sqrt[0.0000301195747116 × 299 792 458 T] Gamma[2] / Gamma[3 / 2] Sqrt[2 / 3]
```

```
87.547590714527 √T
```

Computing time taken to reach surface

```
UnitConvert[Solve[QuantityMagnitude["SolarRadius", "Meters"] == dT, T][[1, 1, 2]] s , "Years"]
```

```
2.0012404104601 × 10^6 yr
```

This matches well with our previous numerical estimate.

Q4)

Importing data from file

```
SetDirectory["C:\\Users\\dan7g\\Google Drive\\Acads\\ASTR404\\"];
dataS = SemanticImport["eker_2014_simple.dat"];
```

Calculating surface gravity and plotting

```
Show[ListLogLogPlot[dataS[All, #], PlotStyle :→ RandomChoice[{Red, Blue}],
    ImageSize → 550, PlotMarkers → "*", Frame → True, FrameLabel →
     {"Mass in solar units", "Gravity at surface in solar units"}, PlotRange → All] & /@
  {<|"M1" → #M1, "g1" → #M1 / #R1^2|> &, <|"M2" → #M2, "g2" → #M2 / #R2^2|> &}]
```



There is a large variance in the surface gravity vs mass. However, it can be seen that higher mass stars generally have a lower surface gravity.

Q3)

Equations

Let M be the total mass of the system, then we have the following equations:

```
eqs =
  {MHI + MHII → X * M,
   MHeI + MHeII + MHeIII → Y * M,
   MHII → xHII (MHI + MHII),
   MHeII → xHeII (MHeI + MHeII + MHeIII),
   MHeIII → xHeIII (MHeI + MHeII + MHeIII),
   Me → MHII me / mp + MHeII me / (4 mp) + 2 MHeIII me / (4 mp)};
```

Mean mass in atomic units

The inverse of the mean mass is given as:

$$\frac{1}{\mu} = \frac{m_p}{M} \sum_{i=1}^{n} \frac{M(i)}{m(i)}$$

Where m(i) is the mass of a particle, and M(i) is the total mass, of the i'th species. In this case we have:

```
eqμ = 1 / μ == mp / M ((MHI + MHII) / mp + (MHeI + MHeII + MHeIII) / (4 mp) + Me / me)
```

$$\frac{1}{\mu} == \frac{\left(\frac{\text{Me}}{\text{me}} + \frac{\text{MHeI}+\text{MHeII}+\text{MHeIII}}{4\,\text{mp}} + \frac{\text{MHI}+\text{MHII}}{\text{mp}}\right)\text{mp}}{\text{M}}$$

Solving for μ

Substituting the first set of equations in the above equation and simplifying, we get:

```
1 / μ == FullSimplify[eqμ[[2]] //. eqs]
```

$$\frac{1}{\mu} == \text{X}\left(1 + \text{xHII}\right) + \frac{1}{4}\left(1 + \text{xHeII} + 2\,\text{xHeIII}\right)\text{Y}$$

QED.

Q2)

Substituting values in Saha equation

```
eqSaha //. {Z2 → 1, Z1 → 2, χ → qM[13.6 eV, "SI"], ne → 6.1 × 10^31, T → 15.7 × 10^6} /. consts
```

$$\frac{\text{NII}}{\text{NI}} == 2.4378887941487$$

For such high temperatures Z1 cannot be taken as 2 since other terms in the series will become significant.

Computing correct value of the Z1 for 20 excited states

```
NSum[2 n^2 Exp[qM[-13.6 eV , "SI"] (1 - 1 / n^2) / (k 1.5 × 10^6)] /. consts, {n, 20}]
5170.5597955066
```

It can be seen that Z1 diverges to infinity for temperatures this high. Also the assumption that electrons do not interact with neighboring nuclei will break down at these temperatures. Therefore the Boltzmann and Saha equations are probably not valid in this regime.

Q1)

Defining constants in SI units

```
qM = QuantityMagnitude;
consts = {h → qM[ h , "J s"], me → qM[ mₑ , "kg"], mp → qM[ mₚ , "kg"], k -> qM[ k , "J/K"]};
```

---

a)

Defining the Saha equation for singly ionized H

```
eqSaha = NII / NI == 2 / ne / Sqrt[h^2 / (2 Pi me k T)]^3 (Z2 / Z1) Exp[-χ / (k T)]
```

$$\frac{\text{NII}}{\text{NI}} == \frac{4\sqrt{2}\, e^{-\frac{\chi}{k T}} \pi^{3/2}\, \text{Z2}}{\text{ne} \left(\frac{h^2}{k\, \text{me}\, T}\right)^{3/2} \text{Z1}}$$

Substituting values for constants

Assuming Z2 = 1, Z1 ~ 2, χ = 13.6 eV, n_e = NII / V, and V = (NI+NII) m_p / ρ and x = NII / (NI+NII)

```
eqS = eqSaha //. {Z2 → 1, Z1 → 2, χ → qM[13.6 eV , "SI"],
      V → (NI + NII) mp / ρ, ne → NII / V, ρ → 10^-6.} /. consts;
```

Rewriting the above equation in terms of ionization fraction x

```
eqx = x (1 - x) # & /@ eqS /. NII → x NI / (1 - x) // Simplify
```

$$x^2 == \frac{1}{\left(\frac{1}{T}\right)^{3/2}} e^{-157\,821.50002145/T} (4.0388497295232 - 4.0388497295232\, x)$$

This is a quadratic equation for the ionization fraction (x).

b)

Plotting ionization fraction vs temperature

```
Plot[Quiet@Solve[eqx && x > 0, x][[1, 1, 2]], {T, 5000, 25 000}, Frame → True, Filling → Bottom,
 ImageSize → 550, FrameLabel → {"Temperature in kelvin", "Ionization fraction"},
 PlotLabel → "Ionization fraction vs temperature for singly ionized Hydrogen"]
```