HW 10 - AstroDynamics

Daniel George

Q2 b)

Reading data from the tabular dump files with B = 0:

```
B0t = Table[Import["A:\\HW10\\B=0\\Blast_B1.00" <>
      If[i < 10, "0", ""] <> ToString[i] <> ".txt", "Data"], {i, 0, 10}];
(* The arrays B0t⟦i⟧ contains the data at t = 0.02(i-1) *)
```

Defining column names

```
iZone = 1; jZone = 2; x1 = 3; x2 = 4; d = 5; V1 = 6;
V2 = 7; V3 = 8; p = 9; B1c = 10; B2c = 11; B3c = 12;
```

Plots

Colormap of pressure at successive time steps



Colormap of density at successive time steps



Radial averages (width of bins = width of zones)

Grouping zones by radial distance

```
grid =
  GatherBy[Tuples[{Range[200], Range[300]}], Floor [Norm [# - {100.5, 150.5}]] &];
```

Averaging density over each bin

```
d0Av = Table[Reverse[Mean /@
    (Extract[Transpose[Partition[B0t[[i]][[All, d]], 200]], #] & /@ grid)], {i, 11}];
```



Averaging radial velocity over each bin

Finding radial component of velocity at each zone:

```
Vr0 = Table[(#[[1]] * #[[3]] + #[[2]] * #[[4]]) / Sqrt[#[[3]]^2 + #[[4]]^2] &[
    Transpose[Partition[B0t[[i]][[All, #]], 200]] & /@ {V1, V2, x1, x2}], {i, 11}];
```

Now averaging it over each radial group of zones:

```
Vr0Av = Table[Reverse[Mean /@ (Extract[Vr0[[i]], #] & /@ grid)], {i, 11}];
```



Averaging pressure over each bin

```
p0Av = Table[Reverse[Mean /@
    (Extract[Transpose[Partition[B0t⟦i⟧⟦All, p⟧, 200]], #] & /@ grid)], {i, 11}];
```



Finding shock front position from pressure profile

We can find the shock front position by calculating the location of the first maxima which we encounter as we move from the edges to the center.

```
sh0 = Transpose[{Range[0, .2, .02],
    .005 ((FindPeaks[#, 0, .05, -∞] & /@ p0Av)⟦All, -1, 1⟧ - 1)}]; sh0⟦1, 2⟧ = 20 * .005;
```

Analytic fit

The Sedov blast is expected to scale as $t^{2/5}$ with time. Therefore we can try fitting this function to our calculated values:

```
fsh0[t_] = Fit[sh0⟦2 ;; -1⟧, {1, t^(2/5)}, t]
```

$-0.07422474526 + 0.9413226992\, t^{2/5}$

Plotting radial position of the shock front vs time



The blue line is the computed value of shock front position. The red line is the analytic fit. We can see that after a small amount of time, the calculated shock front radius matches the analytic fit as expected.

Q2 c)

Reading data from the tabular dump files with B = 1:

```
B1t = Table[Import["A:\\HW10\\B=1\\Blast_B1.00" <>
      If[i < 10, "0", ""] <> ToString[i] <> ".txt", "Data"], {i, 0, 10}];
(* Now the arrays B1t⟦i⟧ contains the data at t = 0.02(i-1) *)
```

Plots

Colormap of pressure at successive time steps



Colormap of density at successive time steps



You can observe that the shock is not spherically symmetric. The rate of expansion of minimum radius of the shock front is comparable to the hydrodynamic case. The maximum and minimum radius of the shock front lies along the x = -y line and the x = y line respectively.

Properties (in cgs units)

Ambient plasma β value

```
Pamb = 0.1;(*Ambient pressure*)
B = 1.;(*Magnetic field strength*)
```

```
β = Pamb / (B^2 / (8 Pi))
```

```
2.513274123
```

Alfven speed

```
ρ0 = 1.;
```

The Alfven speed (Va) is given by:

```
Va = B / Sqrt[4 Pi ρ0]
```

```
0.2820947918
```

Magnetosonic speed

```
Vs = 0.4082482905 ;(*Speed of sound from athinput.blast_B1*)
```

Speeds of the fast and slow MHD waves are functions of angle θ given by the following formulae:

$$\sqrt{\left(\frac{1}{2}\left(\sqrt{(Va^2+Vs^2)^2 \pm 4\, Va^2\, Vs^2 \cos^2\left(\theta-\frac{\pi}{4}\right)}+Va^2+Vs^2\right)\right)}$$

```
VmFast[θ_] :=
  Sqrt[1 / 2 (Va^2 + Vs^2 + Sqrt[(Va^2 + Vs^2)^2 - 4 Va^2 Vs^2 Cos[θ - Pi / 4]^2])];
VmSlow[θ_] := Sqrt[
   1 / 2 (Va^2 + Vs^2 - Sqrt[(Va^2 + Vs^2)^2 - 4 Va^2 Vs^2 Cos[θ - Pi / 4]^2])];
```

Plots of different speeds vs angle



Therefore the speed of the shock will be maximum perpendicular to the magnetic field i.e. along 135° angle (x = -y line) since the dominant fast MHD wave has maximum speed in this direction. Similarly the shock speed along 45° angle (x = y line) will be minimum. Thus the Mach number will be maximum along 135° angle and minimum along 45° angle (see below).

Colormap of Mach number at successive time steps

Mach number at each zone = magnitude of velocity divided by sound speed in that zone:

```
Mach = Table[(#[[1]]^2 + #[[2]]^2) / Sqrt[5 / 3 #[[3]] / #[[4]]] &[
    Transpose[Partition[B1t[[i]][[All, #]], 200]] & /@ {V1, V2, p, d}], {i, 1, 11}];
```



Radial averages (width of bins = width of zones)

Averaging density over each bin

```
d1Av = Table[Reverse[Mean /@
      (Extract[Transpose[Partition[B1t[[i]][[All, d]], 200]], #] & /@ grid)], {i, 11}];
```



Averaging radial velocity over each bin

Finding radial component of velocity at each zone:

```
Vr1 = Table[(#[[1]] * #[[3]] + #[[2]] * #[[4]]) / Sqrt[#[[3]]^2 + #[[4]]^2] &[
    Transpose[Partition[B1t[[i]][[All, #]], 200]] & /@ {V1, V2, x1, x2}], {i, 11}];
```

Now averaging it over each radial group of zones:

```
Vr1Av = Table[Reverse[Mean /@ (Extract[Vr1[[i]], #] & /@ grid)], {i, 11}];
```



Finding maximum and minimum radius of shock front

Since the shock is elliptical and symmetric about the x = y line, we can assume that the maximum and minimum radius of the shock front will be either on the line x = y or on the line x = -y.

Shock front along 45° angle (x = y line)

Finding zones along the 45° line:

```
grid45 = Table[{i, i + 50}, {i, 1, 200}];
```

Finding pressure at different values of radial distance along 45° line:

```
p1Av45 = (#[[All, 100 ;; 1 ;; -1]] + #[[All, 101 ;; 200]]) / 2 &[
    Table[(Extract[Transpose[Partition[B1t[[i]][[All, p]], 200]], grid45]), {i, 11}]];
```



Calculating position of shock front by finding the first peak that we encounter while moving from the edges to the center:

```
sh45 = Transpose[{Range[0, .2, .02], .005 * Sqrt[2]
     ((FindPeaks[#, 0, .05, -∞] & /@ p1Av45) [[All, -1, 1]] - 1)}]; sh45[[1, 2]] = .1;
```

Shock front along 135° angle (x = -y line)

Finding zones along the 135° line:

```
grid135 = Table[{i, 251 - i}, {i, 1, 200}];
```

Finding pressure at different values of radial distance along 135° line:

```
p1Av135 = (#[[All, 100 ;; 1 ;; -1]] + #[[All, 101 ;; 200]]) / 2 &[Table[
     (Extract[Transpose[Partition[B1t[[i]][[All, p]], 200]], grid135]), {i, 11}]];
```



Calculating position of shock front by finding the first peak that we encounter while moving from the edges to the center:

```
sh135 = Transpose[{Range[0, .2, .02], .005 * Sqrt[2]
     ((FindPeaks[#, 0, .05, -∞] & /@ p1Av135) [[All, -1, 1]] - 1)}]; sh135[[1, 2]] = .1;
```

Plots of the maximum and minimum radius of the shock front vs time



It can be observed that the hydrodynamic shock radius (which scales as $t^{2/5}$) closely matches the minimum radius of the MHD shock front. This is because along the magnetic field (45° angle) the fast MHD wave which is dominant has speed equal to the sound speed. Perpendicular to the magnetic field (135° angle) the fast MHD wave speed is greater than the sound speed (see plots of speeds in the previous section) therefore the shock front radius scales faster than the hydrodynamic case.

Q1) a) The MHD equations can be written as:

$$\nabla\cdot B = 0 \; ; \; \partial \bar{B}/\partial t - \nabla\times(\bar{V}\times\bar{B}) = 0$$

$$\partial\rho/\partial t + \nabla\cdot(\rho\bar{V}) = 0 \; ; \; \frac{\partial(\rho\bar{V})}{\partial t} + \nabla\cdot T = 0$$

$$\partial\left(\tfrac{1}{2}\rho V^2 + P/_{\Gamma-1} + B^2/_{2\mu_0}\right)/\partial t$$

$$+ \nabla\cdot\left[\left(\tfrac{1}{2}\rho V^2 + \frac{\Gamma}{\Gamma-1}p\right)V + \frac{\bar{B}\times\bar{V}\times\bar{B}}{\mu_0}\right]$$

(where $T = \rho\bar{V}\bar{V} + (p + B^2/_{2\mu_0})\underline{I} - \bar{B}\bar{B}/\mu_0$)

In the rest frame of the shock with the shock front in the y-z plane:

$\partial/\partial t = \partial/\partial y = \partial/\partial z = 0$; Far from shock $\partial/\partial x = 0$

$\bar{V}_1 = (V_1, 0, 0)$; $\bar{V}_2 = (V_2, 0, 0)$

$\bar{B}_1 = (0, B_1, 0)$; $\bar{B}_2 = (0, B_2, 0)$



Integrating across the shock gives:

$$[V_2B_2 - V_1B_1] = 0$$

$$[\rho_2V_2 - \rho_1V_1] = 0$$

$$[\rho_2V_2^2 + p_2 + B_2^2/2\mu_0 - \rho_1V_1^2 + p_1 + B_1^2/2\mu_0] = 0$$

$$\Big[\tfrac{1}{2}\rho_2V_2^3 + \Gamma/_{\Gamma-1}\, p_2V_2 + \frac{B_2^2V_2}{\mu_0}$$
$$- \tfrac{1}{2}\rho_1V_1^3 + \frac{\Gamma}{\Gamma-1}\, p_1V_1 + \frac{B_1^2V_1}{\mu_0}\Big] = 0$$

These are the Rankine-Hugoniot relations //

Q2) b) From the jump conditions we get

$$B_2/B_1 = \rho_2/\rho_1 = V_1/V_2 = \gamma \rightarrow (1)$$

$$P_2/P_1 = R \quad ; \quad (\text{Let } \beta_1 = 2\mu_0 P_1/B_1^2)$$

where $R = 1 + \Gamma M_1^2(1 - 1/\gamma) + 1/\beta_1 (1-\gamma^2)$,
$M_1 = V_1/V_{S1} = V_1/\sqrt{\Gamma P_1/\rho_1}$,
and γ is a real positive root of the eq:

$$F(\gamma) = 2(2-\Gamma)\gamma^2 + \Gamma(2(1+\beta_1) + (\Gamma-1)\beta_1 M_1^2]\gamma - \Gamma(\Gamma+1)\beta_1 M_1^2 = 0$$

Let γ_1, γ_2 be the two roots

$$\Rightarrow \gamma_1 \times \gamma_2 = \frac{-(\Gamma+1)\Gamma\beta_1 M_1^2}{2(2-\Gamma)}$$

if $\Gamma < 2 \Rightarrow \gamma_1$ or γ_2 is $-$ve

Also it can be seen that the root lies between 0 and $(\Gamma+1)/(\Gamma-1)$ (Since $F(0) < 0$ & $F(\frac{\Gamma+1}{\Gamma-1}) > 0$)

$$\Rightarrow 0 < B_2/B_1 \leq (\Gamma+1)/(\Gamma-1)$$

Solving for +ve γ we get

$$\gamma = B_2/B_1 = \frac{1}{2(4-2\Gamma)}\left[-\beta_1\Gamma^2 M_1^2 + \beta_1\Gamma M_1^2 - 2\beta_1\Gamma - 2\Gamma + \sqrt{(\beta_1\Gamma^2 M_1^2 - \beta_1\Gamma M_1^2 + 2\beta_1\Gamma + 2\Gamma)^2 - 4(4-2\Gamma) \times (-b\Gamma^2 M^2 - b\Gamma M^2)}\right]$$

c) The second law of thermodynamics implies that downstream entropy is greater upstream entropy

$$S = \ln (P/\rho^{\Gamma}) \Rightarrow [S]_1^2 = \ln R - \Gamma \ln \gamma$$

$$[S]_1^2 > 0 \Rightarrow \gamma > 1 \quad (\text{Compressive})$$

$$\text{If } F(1) < 0 \;\&\; F\left(\frac{\Gamma+1}{\Gamma-1}\right) > 0 \Rightarrow 1 < \gamma < \frac{\Gamma+1}{\Gamma-1}$$

$$\Rightarrow F(1) = 2\beta_1 \Gamma (M_1^2 - 1) < 0$$

$$\Rightarrow M_1^2 > 1 + \frac{2}{\Gamma \beta_1}$$